'    Copyright 2008 Jan Molnar <jan.molnar@abds.de>
'
'    This file is part of JCQ.

'    JCQ is free software: you can redistribute it and/or modify
'    it under the terms of the GNU General Public License as published by
'    the Free Software Foundation, either version 2 of the License, or
'    (at your option) any later version.

'    JCQ is distributed in the hope that it will be useful,
'    but WITHOUT ANY WARRANTY; without even the implied warranty of
'    MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.  See the
'    GNU General Public License for more details.

'    You should have received a copy of the GNU General Public License
'    along with JCQ. If not, see <http://www.gnu.org/licenses/>.

Public Class Snac1005
    Inherits Snac

    Public Sub New()
        MyBase.New(&H10, &H5)
    End Sub

    Public Overrides Function CalculateDataSize() As Integer
        Throw New NotImplementedException
    End Function

    Private _Uin As String
    Public Property Uin() As String
        Get
            Return _Uin
        End Get
        Set(ByVal value As String)
            _Uin = value
        End Set
    End Property

    Private _IconHash As New List(Of Byte)
    Public ReadOnly Property IconHash() As List(Of Byte)
        Get
            Return _IconHash
        End Get
    End Property

    Private _IconData As New List(Of Byte)
    Public ReadOnly Property IconData() As List(Of Byte)
        Get
            Return _IconData
        End Get
    End Property

    Public Overrides Sub Deserialize(ByVal data As System.Collections.Generic.List(Of Byte))
        MyBase.Deserialize(data)

        ' xx   byte   uin length 
        ' xx ..   ascii   uin string 

        Dim index As Integer = Snac.SizeFixPart

        _Uin = ByteConverter.ToStringFromByteIndex(index, data)
        index += 1 + _Uin.Length

        ' 00 01   word   icon id (not sure) 
        ' 01   byte   icon flags (bitmask, purpose unknown) 

        index += 3

        ' 10   byte   md5 hash size (16) 
        ' xx ..   array   requested icon md5 hash 

        Dim length As Byte = data(index)
        index += 1

        _IconHash.AddRange(data.GetRange(index, length))
        index += length

        ' xx xx   word   length of the icon 

        Dim iconLength As Integer = ByteConverter.ToUInt16(data.GetRange(index, 2))
        index += 2

        ' xx ..   array   icon data (jfif - jpeg file interchange format) 

        _IconData.AddRange(data.GetRange(index, iconLength))
        index += iconLength

        Me.SetTotalSize(index)
    End Sub

    Public Overrides Function Serialize() As System.Collections.Generic.List(Of Byte)
        Throw New NotImplementedException
    End Function
End Class
